Indian Institute of Technology Bombay

EE 324: Control Systems Lab

Problem Sheet 5 Report

february 21, 2021

Student Name	*Roll Number*
Mantri Krishna Sri Ipsit	180070032

Contents

1 Question 1

1.1 Part a

The closed loop transfer function given is

$$T(s) = \frac{10}{s^3 + 4s^2 + 5s + 10} = \frac{kG}{1 + kG}$$

by solving, we get the plant's open loop transfer function as

$$G(s) = \frac{1}{s^3 + 4s^2 + 5s}$$

The root locus of this plant is found using the following code:

```
// Part a
clc; clear;
//————————————
// Question 1
// Part a
s = poly(0, 's');
Ga = 1/(s^3 + 4*s^2 + 5*s);
Ga = syslin('c', Ga);
scf();
evans(Ga, 30);
```



Figure 1: Root Locus for $G(s) = \frac{1}{s^3+4s^2+5s}$

1.2 Part b

The open loop transfer function given is:

$$G(s) = \frac{s+1}{s^2(s+3.6)}$$

The root locus of this plant is found using the following code:

```
//-----------------------------
// Part b
Gb = (s+1) / (s^2 * (s + 3.6));
Gb = syslin('c', Gb);
scf();
evans(Gb, 80);
```



Figure 2: Root Locus for $G(s) = \frac{s+1}{s^2(s+3.6)}$

1.3 Part c

The open loop transfer function given is:

$$G(s) = \frac{s+0.4}{s^2(s+3.6)}$$

The root locus is plotted using the following code:

```
//-----------------------------
// Part c
Gc = (s+0.4) / (s^2 * (s + 3.6));
Gc = syslin('c', Gc);
scf();
evans(Gc, 50);
```



Figure 3: Root Locus for $G(s) = \frac{s+0.4}{s^2(s+3.6)}$

1.4 Part d

The open loop transfer function given is:

$$G(s) = \frac{s+p}{s(s+1)(s+2)}$$

The root locus for some selected values of p is plotted using the following code:

```
//------------------
// Part d
P = [-5, -2, -1, 0, 1, 2, 3, 4, 5];
scf();
for i=1:size(P, 2)
    p = P(i);
    Gd = (s + p) / (s * (s+1) * (s+2));
    Gd = syslin('c', Gd);
    evans(Gd, 200);
end
title(["Locus of closed loop poles of", "$\frac{s+p}{s(s+1)(s+2)}$", ",",...
"$p \in \{-5,-2,-1,0,1,2,3,4,5\}$"], "fontsize", 3);
```

Please turn over



Figure 4: Root locus for $G(s) = \frac{s+p}{s(s+1)(s+2)}$, $p \in \{-5, -2, -1, 0, 1, 2, 3, 4, 5\}$

If $p < 0$, then we will have real axis segemnts in the ORHP and hence the system will be unstable. If there is pole-zero cancellation, then the system will be stable for all values of proportional gain K_p. For $p = 3$, the root locus has an asymptote on $j\omega$ axis. If $p > 3$, then the non-linear branches of the root locus enter the ORHP and the system will again be unstable for some values of K_p.

2 Question 2

2.1 Part a

The open loop transfer function:

$$G(s) = \frac{s^3 - 1}{s^3 + 1}$$

The root locus is plotted using the following code:

```
//-----------------------
// Question 2
// Part a
s = poly(0, 's');
Ga = (s^3 - 1) / (s^3 + 1);
Ga = syslin('c', Ga);
scf();
evans(Ga, 30);
```

Please turn over



Figure 5: Root Locus of $G(s) = \frac{s^3-1}{s^3+1}$

As we can see, the break-away and break-in points coincide.

2.2 Part b

Consider the following open loop transfer function:

$$G(s) = \frac{s^5 - 1}{s^5 + 1}$$

Its root locus is plotted using the following commands:

```
//-----------------------------------
// Part b
Gb = (s^5 - 1) / (s^5 + 1);
Gb = syslin('c', Gb);
scf();
evans(Gb, 30);
```

We can see in the figure 6 that the number of branches at break-away or break-in point is more than 4.

Please turn over



Figure 6: Root Locus for $G(s) = \frac{s^4-1}{s^4+1}$

2.3 Part c

Consider the following open loop transfer function:

$$G(s) = \frac{1}{s+1}$$

Its root locus is plotted using the following code:

```
//----------------------------
// Part c
Gc = 1 / (s + 1);
Gc = syslin('c', Gc);
scf();
evans(Gc, 50);
```

In the figure 7, we can see that the branch of the root locus coincide with its asymptote.

Please turn over



Figure 7: Root locus for $G(s) = \frac{1}{s+1}$

2.4 Part d

The example that I took is

$$G(s) = \frac{1}{s^4 - 13s^2 + 36}$$

replacing s^2 with $-s^2$, we get

$$G1(s) = \frac{1}{s^4 + 13s^2 + 36}$$

now substituting s with $s - 5$, we get

$$G2(s) = \frac{1}{(s-5)^4 + 13(s-5)^2 + 36}$$

Its root locus is plotted using the following code:

```
//-----------------------
// Part d
Gd = 1 / (s^4 - 13*s^2 + 36);
Gd1 = 1 / (s^4 + 13*s^2 + 36);
Gd1 = syslin('c', Gd1);
Gd2 = 1 / ((s-5)^4 + 13*(s-5)^2 + 36);
Gd2 = syslin('c', Gd2);
scf();
evans(Gd2, 100);
```

We can see in figure 8 that the break-away points are complex numbers.



Figure 8: Root locus for $G2(s) = \frac{1}{(s-5)^4+13(s-5)^2+36}$

3 Question 3

We have the plant's open loop transfer function as:

$$G(s) = \frac{1}{s(s^2+3s+5)}$$

The value of K_p for a rise time of 1.5s will be 3.74. Also, the minimum possible rise time for the given system (maintaining stability) is 0.57s. These were found using the following code:

```
clc; clear;
function [rise_time] = Tr(t, sl, flag)
    outputs = csim('step', t, sl);
    ss_val = mean(outputs(size(outputs, 2)-200:size(outputs, 2)));
    if flag then
        ss_val = 1;
    end
    peak_val = max(outputs);
    rise_time_low = 0;
    rise_time_high = 0;
    for i=1:size(outputs, 2)
        if(outputs(i) - (0.1 * ss_val)>= 5*1e-4)
            rise_time_low = t(i);
            break;
            end
    end
    for i=1:size(outputs, 2)
        if(outputs(i) - 0.9 * ss_val>= 5*1e-4)
            rise_time_high = t(i);
            break;
            end
    end
    rise_time = rise_time_high - rise_time_low;
```

```
endfunction
//-----------------
// Question 3
s = poly(0, 's');
G = 1/(s*(s^2 + 3*s + 5));
K = 0.01:0.01:kpure(G);
t = 0:0.01:20;
scf();
candidates = [];
for i=1:size(K, 2)
    k = K(i);
    T = syslin('c', k*G);
    T = T /. syslin('c', 1, 1);
    tr = Tr(t, T, %f);
    if i == size(K, 2)
        tr = Tr(t, T, %t);
    end
    plot(k, tr, 'b.', 'LineWidth', 0.25);
    if tr == 1.5
        candidates = [candidates, k];
    end
end
xlabel("$K_p$", 'fontsize', 3);
ylabel("Rise Time in seconds", 'fontsize', 3);
title(["Rise time vs ", "$K_p$", "for unity negative feedback system of the plant",...
"$\frac{1}{s(s^2 + 3s + 5)}$"], "fontsize", 3);
disp("Kp for rise time = 1.5s");
disp(candidates(1));
k_critical = kpure(G);
T = syslin('c', k_critical*G);
T = T /. syslin('c', 1, 1);
min_tr = Tr(t, T, %t);
disp("Minimum rise time for stable closed-loop");
disp(min_tr);
```

Please turn over



Figure 9: Rise time vs K_p

4 Question 4

The open loop transfer function is

$$G(s) = \frac{0.11(s+0.6)}{6s^2 + 3.6127s + 0.0572}$$

The steady state error for the step response will be

$$e(\infty) = \frac{1}{1 + K_p G(0)} = 0.01$$

Solving this, will give $K_p = 85.8$. The system will be marginally stable at $K_p = -0.87$. The step response and root locus are as follows:

Please turn over



Figure 10: Step response with $e(\infty) = 0.01$



Figure 11: Root Locus

These were generated using the following code:

```
clc; clear;
//——————————————————
// Question 4
s = poly(0, 's');
g = 0.11 * (s+0.6) / (6*s^2 + 3.6127*s +0.0572);
G = syslin('c', g);
scf();
evans(G, 70);
```

```
scf();
t = 0:0.01:10;
k = 85.8;
plot(t, csim('step', t, (k*G)/.syslin('c', 1, 1)));
K = -1:0.01:-0.1;
for i=1:size(K, 2)
    k = K(i);
    gg = g * k;
    G = syslin('c', gg);
    T = G /. syslin('c', 1, 1);
    [z, p, _p] = tf2zp(T);
    x = real(p);
    if abs(x(1)) <= 1e-4 || abs(x(2)) <= 1e-4
        disp("K at jw crossing");
        disp(k);
    end
end
```

5 Question 5

The example that I had taken is

$$G(s) = \frac{100}{(s+1)(s+2)(s+50)}$$

The root locus of this system will be



Figure 12: Root locus of $G(s) = \frac{100}{(s+1)(s+2)(s+50)}$

The second order approximation of this system will be

$$G1(s) = \frac{2}{(s+1)(s+2)}$$

The root locus of second order approximation is The plot of the closed loop transfer function of the difference



Figure 13: Root locus of $G1(s) = \frac{2}{(s+1)(s+2)}$

of both the plants i.e.

$$T(s) = \frac{k(G - G1)}{1 + k(G - G1)}$$

If G and $G1$ are closer, then the step response will be close to 0. We can see from figure 14 that for $K = 1.2$ the difference between the responses is within 1% i,e., the step responses are similar.

Please turn over



Figure 14: Step response of $T(s) = \frac{k(G-G1)}{1+k(G-G1)}$