Indian Institute of Technology Bombay

EE 324: Control Systems Lab

Problem Sheet 6 Report

March 7, 2021

Student Name	*Roll Number*
Mantri Krishna Sri Ipsit	180070032

Contents

1 Question 1

The open loop transfer function given is

$$G(s) = \frac{1}{(s+3)(s+4)(s+12)}$$

We have to design a proportional controller with gain K using root-locus methods.

1.1 Part a

We need a steady state error of 0.489 on applying a step input. We know that the steady state error is given by

$$e(\infty) = \frac{1}{1 + K\,G(0)} = 0.489$$

on solving this, we get

$$K = 150.47853$$

For designing this controller, we do not need root-locus. Hence, I solved it mathematically.

1.2 Part b

We need to design K such that the damping ratio $\zeta = 0.35$. We know that a constant ζ means %OS is constant and we get a line in the root-locus with slope

$$\tan(\alpha) = -\frac{\sqrt{1-\zeta^2}}{\zeta} = 2.6764277$$

On plotting this on root-locus, we get the desired value of K as

$$K = 371.9$$



Figure 1: Damping ratio = 0.35

1.3 Part c

The gain value at the break away point is

$$K = 2.127$$

This was found using the root-locus.



Figure 2: Gain at break away point

1.4 Part d

Here the gain K is varied in a small range from 0 to 1. The step responses and the steady state errors vary as follows: We can make the following observations:

- For this small range of K, the closed loop poles lie on the real axis and are between the dominant open-loop poles.
- As the value of K is increased, the steady-state error is decreased. This is because the steady state error is given by

  $$e(\infty) = \frac{1}{1 + K/144}$$

  This is essentially a hyperbola, but for small values of K, it looks like a straight line.
- As the value of K is increased, the rise time and settling times decrease, as visible from the graph.



Figure 3: Step responses as K is increased from 0 to 1



Figure 4: Steady State Error as K is increased from 0 to 1



Figure 5: Portion of root-locus as K is increased from 0 to 1

1.5 Part e

Here we vary K from 1 to 1000 i.e., over a large range. We observe the step responses as follows:



Figure 6: Step responses as K is increased from 1 to 1000



Figure 7: Steady State Error as K is increased from 1 to 1000



Figure 8: Portion of root-locus as K is increased from 1 to 1000

We can make the following observations:

- The closed-loop poles lie on the real axis for smaller values of K and then break-away and more towards the imaginary axis. It is due to this reason, the step response is initially looks like an overdamped response and settling time decreases and after crossing the break away point, the step response becomes oscillatory and looks like an underdamped response with settling time increasing.
- As we observe from the plot, the steady state error decreases in the form of a hyperbola as expected. This is because the plant has a Type 0 transfer function.
- From the step response plots, we can say that as K increases, the response takes more time to settle as it undergoes more oscillations. This is because the closed loop poles move towards the imaginary axis.
- If we observe the root-locus, we can see that the closed loop poles have not entered the open right half plane for this range of the gain K. Hence the system will be stable.

All the above figures have been generated using the following code:

```
clc; clear;
s = poly(0, 's');
g = 1 / ((s+3)*(s+4)*(s+12));
G = syslin('c', g);
//----------------------------
// Part a
Ka = ((1/0.489)-1)*144;
disp("K for SSE of 0.489 = ");
disp(Ka);
//----------------------------
// Part b
scf();
evans(G, 700);
x = -2.3:0.00001:0;
ratio = sqrt(1-0.35^2) / 0.35;
y = -ratio .* x;
plot(x, y, 'r-', 'LineWidth', 1);
L = legend(['open-loop poles', 'asymptotes', 'branch 3', 'branch 2', 'branch 1',...
"$\zeta = 0.35$"]);
L.font_size = 3;
L.legend_location = "in_upper_left";
title(["Designing P controller for a damping ratio of 0.35"], 'fontsize', 3);
poi = 2.061 + %i*5.516; // from the root locus
Kb = 371.9; // from the root locus
disp("K for damping ratio of 0.35 = ");
disp(Kb);
// --------------------------
// Part c
Kc = 2.127 // from the root locus
disp("Gain value at the break away point");
disp(Kc);
scf();
evans(G, 700);
// --------------------------
// Part d
K = %eps:0.1:1;
t = 0:0.001:10;
errs = [];
scf();
for i=1:size(K, 2)
    k = K(i);
    G = syslin('c', k*g);
    T = G /. syslin('c', 1, 1);
    xset("thickness", 2);
    plot(t, csim('step', t, T));
    xgrid;
    sse = 1 / (1 + k / (3 * 4 * 12));
    errs = [errs, sse];
end
xlabel("$t$", 'fontsize', 3);
ylabel("Step response", 'fontsize', 3);
title(["Unity gain negative feedback for ", "$\frac{K}{(s+3)(s+4)(s+12)}$"],...
```

```
"fontsize", 3);
scf();
evans(syslin('c', g), 1);
scf();
plot(K, errs, 'b-', 'LineWidth', 2);
xlabel("$K$", "fontsize", 3);
ylabel("Steady State Error");
title(["Unity gain negative feedback for ", "$\frac{K}{(s+3)(s+4)(s+12)}$"],...
"fontsize", 3);
// --------------------------------
// Part e
K = 1:50:1000;
t = 0:0.001:10;
errs = [];
scf();
for i=1:size(K, 2)
    k = K(i);
    G = syslin('c', k*g);
    T = G /. syslin('c', 1, 1);
    xset("thickness", 2);
    plot(t, csim('step', t, T));
    xgrid;
    sse = 1 / (1 + k / (3 * 4 * 12));
    errs = [errs, sse];
end
xlabel("$t$", 'fontsize', 3);
ylabel("Step response", 'fontsize', 3);
title(["Unity gain negative feedback for ", "$\frac{K}{(s+3)(s+4)(s+12)}$"],...
"fontsize", 3);
scf();
evans(syslin('c', g), 1000);
scf();
plot(K, errs, 'b-', 'LineWidth', 2);
xlabel("$K$", "fontsize", 3);
ylabel("Steady State Error");
title(["Unity gain negative feedback for ", "$\frac{K}{(s+3)(s+4)(s+12)}$"],...
"fontsize", 3);
```